package XComponent.show
{
	/**
	 * 应scalex9只支持矢量图故有此
	 * 原理很简单,
	 * 用程序把位图切成九块,
	 * 装里一个Sprite里,
	 * 然后重写Sprite的width和height这两个方法,
	 * 根据改变大小来重新设置位图的各个位置…这样就实现了位图的不变形缩放….
	 */	
	import flash.display.Bitmap;
	import flash.display.BitmapData;
	import flash.display.DisplayObject;
	import flash.display.Sprite;
	import flash.geom.Point;
	import flash.geom.Rectangle;
	
	public class XBitmapScale9Grid		extends Sprite
	{	
		private var source : Bitmap;
		
		private var scaleGridTop : Number;
		
		private var scaleGridBottom : Number;
		
		private var scaleGridLeft : Number;
		
		private var scaleGridRight : Number;
		
		
		private var leftUp : Bitmap;
		
		private var leftCenter : Bitmap;
		
		private var leftBottom : Bitmap;
		
		private var centerUp : Bitmap;
		
		private var center : Bitmap;
		
		private var centerBottom : Bitmap;
		
		private var rightUp : Bitmap;
		
		private var rightCenter : Bitmap;
		
		private var rightBottom : Bitmap;
		
		
		
		private var _width : Number;
		
		private var _height : Number;
		
		
		private var minWidth : Number;
		
		private var minHeight : Number;
		
		/**
		 * 
		 * @param source  如果是bitmap Ok不是就要创建一个bitmap
		 * @param scaleGridTop
		 * @param scaleGridBottom
		 * @param scaleGridLeft
		 * @param scaleGridRight
		 * 
		 */		
		public function XBitmapScale9Grid(source:Bitmap=null, scaleGridTop:Number=0, scaleGridBottom:Number=0, scaleGridLeft:Number=0, scaleGridRight:Number=0)
		{
			if(!source) return;
			
			Init(source,scaleGridTop,scaleGridBottom,scaleGridLeft,scaleGridRight);
		}
		
		public function Init(source:Bitmap, scaleGridTop:Number, scaleGridBottom:Number, scaleGridLeft:Number, scaleGridRight:Number) : void
		{
			clearChilderen();
			
			if(source is Bitmap)
			{
				this.source = source as Bitmap;
			}
			else
			{
				var bitD:BitmapData = new BitmapData(source.width,source.height,true);
				bitD.draw(source);
				var bit : Bitmap = new Bitmap(bitD);
				
				this.source = bit;
			}
			
			this.scaleGridTop = scaleGridTop;
			
			this.scaleGridBottom = scaleGridBottom;
			
			this.scaleGridLeft = scaleGridLeft;
			
			this.scaleGridRight = scaleGridRight;
			
			init();
		}
		
		private function init() : void {
			
			_width = source.width;
			
			_height = source.height;
			
			
			
			leftUp = getBitmap(0, 0, scaleGridLeft, scaleGridTop);
			
			this.addChild(leftUp);
			
			
			
			leftCenter = getBitmap(0, scaleGridTop, scaleGridLeft, scaleGridBottom - scaleGridTop);
			
			this.addChild(leftCenter);
			
			
			
			leftBottom = getBitmap(0, scaleGridBottom, scaleGridLeft, source.height - scaleGridBottom);
			
			this.addChild(leftBottom);
			
			
			
			centerUp = getBitmap(scaleGridLeft, 0, scaleGridRight - scaleGridLeft, scaleGridTop);
			
			this.addChild(centerUp);
			
			
			
			center = getBitmap(scaleGridLeft, scaleGridTop, scaleGridRight - scaleGridLeft, scaleGridBottom - scaleGridTop);
			
			this.addChild(center);
			
			
			
			centerBottom = getBitmap(scaleGridLeft, scaleGridBottom, scaleGridRight - scaleGridLeft, source.height - scaleGridBottom);
			
			this.addChild(centerBottom);
			
			
			
			rightUp = getBitmap(scaleGridRight, 0, source.width - scaleGridRight, scaleGridTop);
			
			this.addChild(rightUp);
			
			
			
			rightCenter = getBitmap(scaleGridRight, scaleGridTop, source.width - scaleGridRight, scaleGridBottom - scaleGridTop);
			
			this.addChild(rightCenter);
			
			
			
			rightBottom = getBitmap(scaleGridRight, scaleGridBottom, source.width - scaleGridRight, source.height - scaleGridBottom);
			
			this.addChild(rightBottom);
			
			
			
			minWidth = leftUp.width + rightBottom.width;
			
			minHeight = leftBottom.height + rightBottom.height;
			
		}
		
		
		
		private function getBitmap(x:Number, y:Number, w:Number, h:Number) : Bitmap {
			
			var bit:BitmapData = new BitmapData(w, h);
			
			bit.copyPixels(source.bitmapData, new Rectangle(x, y, w, h), new Point(0, 0));
			
			var bitMap:Bitmap = new Bitmap(bit);
			
			bitMap.x = x;
			
			bitMap.y = y;
			
			return bitMap;
			
		}
		
		
		
		override public function set width(w : Number) : void {
			
			if(w < minWidth) {
				
				w = minWidth;
				
			}
			
			_width = w;
			
			refurbishSize();
			
		}
		
		override public function get width() : Number
		{
			return _width;
		}
		
		override public function set height(h : Number) : void {
			
			if(h < minHeight) {
				
				h = minHeight;
				
			}
			
			_height = h;
			
			refurbishSize();
			
		}
		
		override public function get height() : Number
		{
			return _height;
		}
		
		
		
		private function refurbishSize() : void {
			
			leftCenter.height = _height - leftUp.height - leftBottom.height;
			
			leftBottom.y = _height - leftBottom.height;
			
			centerUp.width = _width - leftUp.width - rightUp.width;
			
			center.width = centerUp.width;
			
			center.height = leftCenter.height;
			
			centerBottom.width = center.width;
			
			centerBottom.y = leftBottom.y;
			
			rightUp.x = _width - rightUp.width;
			
			rightCenter.x = rightUp.x;
			
			rightCenter.height = center.height;
			
			rightBottom.x = rightUp.x;
			
			rightBottom.y = leftBottom.y;
			
		}
		
		private function clearChilderen() : void
		{
			for(var i:int = this.numChildren-1;i>=0;i--)
			{
				this.removeChildAt(i);
			}
		}
	}
}